Metódy Runge-Kutta

Metódy Runge-Kutta sú jednokrokové metódy vyšších rádov, ktoré je možné odvodiť podobným spôsobom ako Eulerovu metódu pre riešenie začiatočných úloh $y' = f(t, y)$, $y(t_0) = y_0$. Počas výpočtu nie je potrebné počítať derivácie vyšších rádov ako napríklad pri Taylorových metódach, čo výrazným spôsobom umožňuje urýchliť výpočet rovnice.
Z teórie vieme, že všeobecný výraz pre explicitné metódy Runge-Kutta je

$$y_{n+1} = y_n + h \sum_{i=1}^{s} b_i k_i$$

$$k_i = f\left(t_n + c_i h,\ y_n + h \sum_{j=1}^{s} a_{ij} k_j\right)$$

kde $a_{ij} = 0$ to $j \geq i$ and $\sum_{j=1}^{s} a_{ij} = c_i$.

Klasická metóda Runge-Kutta druhého rádu s dvoma stupňami má nasledujúcu Butcherovu tabuľku:

$$\begin{array}{c|c|c} 0 & 0 & . \\ \hline c_2 & c_2 & 0 \\ \hline . & b_1 & b_2 \end{array}$$

kde koeficienty diagramu overuje systém rovníc:

$$b_1 + b_2 = 1$$
$$b_2 c_2 = \tfrac{1}{2}$$

Existuje preto nekonečne veľa metód Runge-Kutta druhého rádu. Najčastejšie sa používajú:

a) **Eulerova modifikovaná metóda**, ktorá zodpovedá voľbe koeficientov $b_1 = 0$, $b_2 = 1$, $c_2 = \frac{1}{2}$. Jej iteračná schéma má tvar $t_{n+1} = t_n + h$, $y_{n+1} = y_n + h k_2$, kde $k_1 = f(t_n, y_n)$ a $k_2 = f\left(t_n + \frac{h}{2}, y_n + \frac{h k_1}{2}\right)$.

V systéme *Mathematica* naprogramujeme dve procedúry, **eulermod** a **eulermodgraf**. Tieto procedúry nám umožnia vykonať výpočty tabuliek hodnôt a nakresliť grafy približných riešení získaných modifikovanou Eulerovou metódou:

```
eulermod [f_,  h_,  ini_, a_, b_] :=
Module [ {yrk, t, y, rktable1, c},
 c = (b - a) / h;
 yrk[0] = ini;
 t[n_] := a + n h;
 yrk[n_] :=
  Module[{k1, k2},
   k1 = f[t[n - 1], yrk[n - 1]];
   k2 = f[t[n - 1] + h/2, yrk[n - 1] + h/2 k1];
   yrk[n] = yrk[n - 1]  + h k2];
 rktable1 = Table[yrk[i], {i, 0, c}];
Table[{t[i], rktable1[[i + 1]]}, {i, 0, c}] // TableForm]
```

```
eulermodgraf[f_, h_, ini_, a_, b_] := Module[{yrk, t, y, rktable1, c}, c = (b - a)/h;
  yrk[0] = ini; t[n_] := a + n h; yrk[n_] := Module[{k1, k2}, k1 = f[t[n - 1], yrk[n - 1]];
    k2 = f[t[n - 1] + h/2, yrk[n - 1] + (h k1)/2]; yrk[n] = yrk[n - 1] + h k2];
  rktable1 = Table[yrk[i], {i, 0, c}]; ListPlot[Table[{t[i], rktable1[[i + 1]]}, {i, 0, c}],
   Joined → True, PlotStyle → {RGBColor[1, 0, 0]}, PlotRange → All];
  Print["y[", t[c], "]=", rktable1[[c + 1]]]]
```

kde **f** je funkcia zodpovedajúca diferenciálnej rovnici, **h** je dĺžka kroku, **ini** je hodnota začiatočnej podmienky a **a** a **b** sú hranice intervalu.

b) **Heunova metóda**, ktorá zodpovedá voľbe koeficientov $b_1 = \frac{1}{2}$, $b_2 = \frac{1}{2}$, $c_2 = 1$. Jej iteračná schéma má tvar $t_{n+1} = t_n + h$, $y_{n+1} = y_n + \frac{h(k_1+k_2)}{2}$ s $k_1 = f(t_n, y_n)$ a $k_2 = f(t_n + h, y_n + h k_1)$.

V systéme *Mathematica* naprogramujeme dve procedúry, **mejoreuler** a **mejoreulergraf**. Tieto vypočítajú tabuľku hodnôt a poskytnú grafické riešenie použitej Heunovej metódy:

```
mejoreuler [f_,  h_,  ini_, a_, b_] :=
Module [ {yrk, t, y, rktable1, c},
 c = (b - a) / h;
 yrk[0] = ini;
 t[n_] := a + n h;
 yrk[n_] :=
  Module[{k1, k2},
   k1 = f[t[n - 1], yrk[n - 1]];
   k2 = f[t[n - 1] + h, yrk[n - 1] + h k1];
   yrk[n] = yrk[n - 1]  + (h / 2) (k1 + k2)];
 rktable1 = Table[yrk[i], {i, 0, c}];
Table[{t[i], rktable1[[i + 1]]}, {i, 0, c}] // TableForm]
```

```
mejoreulergraf[f_, h_, ini_, a_, b_] := Module[{yrk, t, y, rktable1, c}, c = (b - a)/h;
  yrk[0] = ini; t[n_] := a + n h; yrk[n_] := Module[{k1, k2}, k1 = f[t[n - 1], yrk[n - 1]];
    k2 = f[t[n - 1] + h, yrk[n - 1] + h k1]; yrk[n] = yrk[n - 1] + 1/2 h (k1 + k2)];
  rktable1 = Table[yrk[i], {i, 0, c}]; ListPlot[Table[{t[i], rktable1[[i + 1]]}, {i, 0, c}],
   Joined -> True, PlotStyle -> {RGBColor[1, 0, 0]}, PlotRange -> All];
  Print["y[", t[c], "]=", rktable1[[c + 1]]]]
```

kde **f** je funkcia zodpovedajúca diferenciálnej rovnici, **h** je dĺžka kroku, **ini** je hodnota začiatočnej podmienky a **a** a **b** sú hranice intervalu.

Obidve metódy je možné použiť na nájdenie riešenia začiatočnej úlohy $y' = -t\,y + \frac{4t}{y}$, $y(0)=1$ na intervale [0,1], s dĺžkou kroku 0.1.

```
f[t_, y_] = -t y + (4 t)/y;

eulermod[f, 0.1, 1, 0, 1]
```

0	1
0.1	1.015
0.2	1.05783
0.3	1.12286
0.4	1.20303
0.5	1.29151
0.6	1.38258
0.7	1.47185
0.8	1.55615
0.9	1.63337
1.	1.70225

```
eulermodgraf[f, 0.1, 1, 0, 1]
```



y[1.]=1.70225

```
mejoreuler[f, 0.1, 1, 0, 1]
```

```
0       1
0.1     1.015
0.2     1.05749
0.3     1.12202
0.4     1.20169
0.5     1.28977
0.6     1.38058
0.7     1.46972
0.8     1.55398
0.9     1.63123
1.      1.70021
```

```
mejoreulergraf[f, 0.1, 1, 0, 1]
```




```
y[1.]=1.70021
```

V predchádzajúcej kapitole 9 - opisujúcej jednokrokovú Eulerovu metódu bolo ukázané, že presné riešenie začiatočnej úlohy pre $t = 1$ je 1.70187. Preto v tomto prípade riešenia, ktoré získame pomocou modifikácie tejto metódy, sú lepšie ako aproximácia získaná pôvodným postupom. Vypovedá o tom aj rád metódy.

Klasická metóda Runge-Kutta tretieho rádu s tromi stupňami má nasledujúcu Butcherovu tabuľku

0	0	.	.
c_2	c_2	0	.
c_3	$c_3 - a_{32}$	a_{32}	0
.	b_1	b_2	b_3

kde koeficienty iteračnej schémy musia spĺňať sústavu rovníc:

$$b_1 + b_2 + b_3 = 1$$

$$b_2\, c_2 + b_3\, c_3 = \frac{1}{2}$$

$$b_2\, c_2^2 + b_3\, c_3^2 = \frac{1}{3}$$

$$b_3\, c_2\, a_{32} = \frac{1}{6}$$

Rovnice sú závislé, preto existuje nekonečný počet metód Runge-Kutta tretieho rádu. Najpoužívanejšia je metóda:

0	0	.	.
$\frac{1}{2}$	$\frac{1}{2}$	0	.
1	-1	2	0
.	$\frac{1}{6}$	$\frac{2}{3}$	$\frac{1}{6}$

Jej iteračná schéma má tvar $t_{n+1} = t_n + h$, $y_{n+1} = y_n + \frac{h(k_1+4k_2+k_3)}{6}$, kde $k_1 = f(t_n, y_n)$, $k_2 = f\left(t_n + \frac{h}{2}, y_n + \frac{h k_1}{2}\right)$ a $k_3 = f(t_n + h, y_n + 2 h k_2 - h k_1)$.

V systéme *Mathematica* naprogramujeme dve procedúry **Runge3** a **Runge3graf**. Tieto vypočítajú tabuľku hodnôt a poskytnú grafickú reprezentáciu riešenia pomocou metódy Runge-Kutta tretieho rádu:

```
Runge3[f_, h_, ini_, a_, b_] :=
Module[{yrk3, t, rktable3, c},
 c = (b - a) / h;
 yrk3[0] = ini;
 t[n_] := a + n h;
 yrk3[n_] :=
  Module[{k1, k2, k3},
   k1 = f[t[n - 1], yrk3[n - 1]];
   k2 = f[t[n - 1] + h / 2, yrk3[n - 1] + (h / 2) k1];
   k3 = f[t[n - 1] + h, yrk3[n - 1] - h k1 + 2 h k2];
  yrk3[n] = yrk3[n - 1] + h ((1 / 6) k1 + (2 / 3) k2 + (1 / 6) k3)];
 rktable3 = Table[yrk3[i], {i, 0, c}];
Table[{t[i], rktable3[[i + 1]]}, {i, 0, c}] // TableForm]
```

```
Runge3graf[f_, h_, ini_, a_, b_] :=
 Module[{yrk3, t, rktable3, c}, c = (b - a)/h; yrk3[0] = ini; t[n_] := a + n h; yrk3[n_] :=
   Module[{k1, k2, k3}, k1 = f[t[n - 1], yrk3[n - 1]]; k2 = f[t[n - 1] + h/2, yrk3[n - 1] + h k1/2];
    k3 = f[t[n - 1] + h, yrk3[n - 1] - h k1 + 2 h k2]; yrk3[n] = yrk3[n - 1] + h (k1/6 + 2 k2/3 + k3/6)];
  rktable3 = Table[yrk3[i], {i, 0, c}]; ListPlot[Table[{t[i], rktable3[[i + 1]]}, {i, 0, c}],
   Joined -> True, PlotStyle -> {RGBColor[1, 0, 0]}, PlotRange -> All];
  Print["y[", t[c], "]=", rktable3[[c + 1]]]]
```

kde **f** je funkcia zodpovedajúca diferenciálnej rovnici, **h** je dĺžka kroku, **ini** je hodnota začiatočnej podmienky a **a** a **b** sú hranice intervalu.

Obidve metódy sa použijú na vyriešenie začiatočnej úlohy $y' = -t\,y + \frac{4t}{y}$, *y(0)=1* na intervale [0,1] s dĺžkou kroku 0,1.

```
f[t_, y_] = -t y + 4 t/y;
```

```
Runge3[f, 0.1, 1, 0, 1]
```

```
0       1
0.1     1.01476
0.2     1.05708
0.3     1.12157
0.4     1.20135
0.5     1.28967
0.6     1.38082
0.7     1.47033
0.8     1.55497
0.9     1.63259
1.      1.70187
```

```
Runge3graf[f, 0.1, 1, 0, 1]
```




```
y[1.]=1.70187
```

Najčastejšie používaná metóda je Runge-Kutta metóda štvrtého rádu, ktorá je reprezentovaná Butcherovou tabuľkou:

0	0			
$\frac{1}{2}$	$\frac{1}{2}$	0		
$\frac{1}{2}$	0	$\frac{1}{2}$	0	
1	0	0	1	0
	$\frac{1}{6}$	$\frac{1}{3}$	$\frac{1}{3}$	$\frac{1}{6}$

a jej iteračná schéma má tvar

$$y_{n+1} = y_n + \frac{h}{6}\,(K_1 + 2\,K_2 + 2\,K_3 + K_4)$$
$$K_1 = f(t_n,\, y_n)$$
$$K_2 = f\left(t_n + \frac{h}{2},\, y_n + \frac{h}{2}\,K_1\right)$$
$$K_3 = f\left(t_n + \frac{h}{2},\, y_n + \frac{h}{2}\,K_2\right)$$
$$K_4 = f(t_n + h,\, y_n + h\,K_3)$$
$$t_{n+1} = t_n + h\,.$$

Znovu v systéme *Mathematica* naprogramujeme procedúry **Runge4** a **Runge4graf** . Tieto vypočítajú tabuľku hodnôt a poskytnú grafickú reprezentáciu riešenia pomocou metódy Runge-Kutta štvrtého rádu:

```
Runge4 [f_, h_, ini_, a_, b_] :=
Module [ {yrk4, t, rktable4, c },
 c = (b - a) / h;
 yrk4[0] = ini;
 t[n_] := a + n h;
 yrk4[n_] :=
  Module[{k1, k2, k3, k4 },
   k1 = f[t[n - 1], yrk4[n - 1]];
   k2 = f[t[n - 1] + h / 2, yrk4[n - 1] + (h / 2) k1];
   k3 = f[t[n - 1] + h / 2, yrk4[n - 1] + (h / 2) k2];
   k4 = f[t[n - 1] + h, yrk4[n - 1] + h k3];
   yrk4[n] =
    yrk4[n - 1] +
     (h / 6) (k1 + 2 k2 + 2 k3 + k4)];
 rktable4 = Table[yrk4[i], {i, 0, c}];
Table[{t[i], rktable4[[i + 1]]}, {i, 0, c}] // TableForm]
```

```
Runge4graf[f_, h_, ini_, a_, b_] := Module[{yrk4, t, rktable4, c}, c = (b - a)/h; yrk4[0] = ini;
  t[n_] := a + n h; yrk4[n_] := Module[{k1, k2, k3, k4}, k1 = f[t[n - 1], yrk4[n - 1]];
    k2 = f[t[n - 1] + h/2, yrk4[n - 1] + (h k1)/2]; k3 = f[t[n - 1] + h/2, yrk4[n - 1] + (h k2)/2];
    k4 = f[t[n - 1] + h, yrk4[n - 1] + h k3]; yrk4[n] = yrk4[n - 1] + 1/6 h (k1 + 2 k2 + 2 k3 + k4)];
  rktable4 = Table[yrk4[i], {i, 0, c}]; ListPlot[Table[{t[i], rktable4[[i + 1]]}, {i, 0, c}],
   Joined -> True, PlotStyle -> {RGBColor[1, 0, 0]}, PlotRange -> All];
  Print["y[", t[c], "]=", rktable4[[c + 1]]]]
```

kde **f** je funkcia zodpovedajúca diferenciálnej rovnici, **h** je dĺžka kroku, **ini** je hodnota začiatočnej podmienky a **a** a **b** sú hranice intervalu.

Obidve metódy použijeme na nájdenie riešenia začiatočnej úlohy $y' = -t\,y + \frac{4t}{y}$, $y(0)=1$ na intervale $[0,1]$, s dĺžkou kroku 0,1.

```
f[t_, y_] = -t y + 4 t / y;
```

```
Runge4[f, 0.1, 1, 0, 1]
```

0	1
0.1	1.01482
0.2	1.05718
0.3	1.1217
0.4	1.20149
0.5	1.28981
0.6	1.38093
0.7	1.47042
0.8	1.55503
0.9	1.63261
1.	1.70187

```
Runge4graf[f, 0.1, 1, 0, 1]
```



y[1.]=1.70187

Príklad 1.

Použime metódu Runge-Kutta štvrtého rádu na nájdenie riešenia začiatočnej úlohy $y' = \frac{y^2-3t^2-2ty}{t^2+2ty}$, *y(1)=2, na intervale [1,2] s krokom h=0.1.*

Riešenie

Definujme funkciu zodpovedajúcu pravej strane rovnice v zadanej začiatočnej úlohe a použime pripravené procedúry na nájdenie riešenia.

$$f[t\_, y\_] := \frac{y^2 - 3\,t^2 - 2\,t\,y}{t^2 + 2\,t\,y}$$

```
Runge4[f, 0.1, 2, 1, 2]
```

1	2
1.1	1.93191
1.2	1.84842
1.3	1.75041
1.4	1.63842
1.5	1.5127
1.6	1.37319
1.7	1.21949
1.8	1.05082
1.9	0.865842
2.	0.662386

```
Runge4graf[f, 0.1, 2, 1, 2]
```



y[2.]=0.662386

Príklad 2.

Riešte začiatočnú úlohu $y' = \sqrt{y} - \frac{20\, e^{-100\,(t-2)^2}}{\sqrt{\pi}}$, $y(1) = 1$. *Nájdite približnú hodnotu riešenia v bode t=3, použitím procedúry Runge4 a s krokom dĺžky h=0.01. Urobte analýzu grafu a porovnajte ju so začiatočnou úlohou.*

Riešenie

Pri pokuse o nájdenie presného riešenia začiatočnej úlohy vidíme, že systém ***Mathematica*** túto začiatočnú úlohu nedokáže vyriešiť:

DSolve$\left[\left\{y'[t] == \sqrt{y[t]} - \frac{20\, e^{-100\,(t-2)^2}}{\sqrt{\pi}},\ y[1] == 1\right\},\ y[t],\ t\right]$

```
Solve::ifun :
 Inverse functions are being used by Solve, so some solutions may not be found;
   use Reduce for complete solution information. More…
```

DSolve$\left[\left\{y'[t] == -\frac{20\, e^{-100\,(-2+t)^2}}{\sqrt{\pi}} + \sqrt{y[t]},\ y[1] == 1\right\},\ y[t],\ t\right]$

Definujme funkciu zodpovedajúcu pravej strane začiatočnej úlohy a aplikujeme metódu Runge-Kutta 4. rádu

f[t_, y_] := $\sqrt{y} - \frac{20\, e^{-100\,(t-2)^2}}{\sqrt{\pi}}$

Runge4graf[f, 0.01, 1, 1, 3]




```
y[3.]=1.03349
```

Teraz vyriešime pomocou rovnakej procedúry aj druhú začiatočnú úlohu.

f[t_, y_] := $\sqrt{y}$

Runge4graf[f, 0.01, 1, 1, 3]




```
y[3.]=4.
```

Grafy sa približne zhodujú až do hodnoty $t = 2$. Potom riešenie druhej začiatočnej úlohy začne rýchlo divergovať. Náhly pokles v grafe riešenia prvej začiatočnej úlohy je spôsobený impulzom z pravej strany rovnice, ktorý zmení správanie riešení tejto rovnice. Impulz môžeme nakresliť nasledovne:

$$\textbf{Plot}\left[\frac{20\, e^{-100\,(t-2)^2}}{\sqrt{\pi}}, \{t, 1, 4\}, \text{PlotRange} \to \text{All}\right]$$